

2022. Лекция 3, законы сохранения КдФ

Функции, определяющие D_x , $g_*(f) = \nabla_f(g)$, вариационную производную и интегрирование по частям

in[1]:=

```
vars[f_] := Union[Cases[f, _u, {0, ∞}]]
dif[f_] := Plus@@(D[f, #] dd[#] & /@ vars[f])
dx[f_] := D[f, x] + dif[f] /. dd[u[k_]] => u[k+1]
dx[f_, k_] := Nest[dx, f, k]
star[g_, f_] := dif[g] /. dd[u[k_]] => dx[f, k]
vard[f_] := Expand[Plus@@((-1)^#[[1]] dx[D[f, #], #[[1]]] & /@ vars[f])]

(* порядок по производным *)
ord[f_] := Module[{v = vars[f]}, If[v == {}, -1, v[[-1]][[1]]]]

int[f_] := Module[{a = 0, b = Expand[f], c, k},
  While[k = ord[b]; (* вычисляем k - порядок b *)
    And[k > 0, (* выполняем цикл, пока выполняется условие, что k>0 и *)
      c = D[b, u[k]]; (* b линейна по старшей переменной *)
      Expand[D[c, u[k]]] == 0
    ],
    c = Integrate[c, u[k-1]]; (* интегрируем по переменной, следующей по старшинству *)
    a = Expand[a + c]; (* пересчет a и b *)
    b = Expand[b - dx[c]]
  ];
  If[ord[b] == -1, (* если в результате остаток не зависит от u, то это функция от x *)
    {Expand[a + Integrate[b, x]], 0},
    {a, b}
  ]
]
```

Первые четыре плотности

Проверим простейшие законы сохранения. Функция $\text{int}[f]$ возвращает пару $\{a, b\}$ такую, что $f = D_x(a) + b$. Если $b = 0$, значит f полная производная.

```
In[9]:= ut = u[3] + 6 u[0] u[1];

Expand[star[u[0], ut]];
int[%]

Expand[star[u[0]^2, ut]];
int[%]

Expand[star[u[1]^2 - 2 u[0]^3, ut]];
int[%]

Expand[star[u[2]^2 - 10 u[0] u[1]^2 + 5 u[0]^4, ut]];
int[%]
```

Out[11]= $\{3 u[0]^2 + u[2], 0\}$

Out[13]= $\{4 u[0]^3 - u[1]^2 + 2 u[0] u[2], 0\}$

Out[15]= $\{-9 u[0]^4 + 12 u[0] u[1]^2 - 6 u[0]^2 u[2] - u[2]^2 + 2 u[1] u[3], 0\}$

Out[17]= $\{24 u[0]^5 - 90 u[0]^2 u[1]^2 + 20 u[0]^3 u[2] + 10 u[1]^2 u[2] + 16 u[0] u[2]^2 - 20 u[0] u[1] u[3] - u[3]^2 + 2 u[2] u[4], 0\}$

```
In[18]:= Expand[-star[u[0], ut] + dx[u[2] + 3 u[0]^2]]

Expand[-star[u[0]^2, ut] + dx[2 u[0] u[2] - u[1]^2 + 4 u[0]^3]]

Expand[
  -star[u[1]^2 - 2 u[0]^3, ut] + dx[2 u[1] u[3] - u[2]^2 - 6 u[0]^2 u[2] + 12 u[0] u[1]^2 - 9 u[0]^4]]

Expand[-star[u[2]^2 - 10 u[0] u[1]^2 + 5 u[0]^4, ut] + dx[2 u[2] u[4] - u[3]^2 - 20 u[0] u[1] u[3] +
  16 u[0] u[2]^2 + 10 u[1]^2 u[2] + 20 u[0]^3 u[2] - 90 u[0]^2 u[1]^2 + 24 u[0]^5]]]
```

Out[18]= 0

Out[19]= 0

Out[20]= 0

Out[21]= 0

Преобразование Миуры

Рассмотрим модифицированное уравнение КдФ, точнее, мКдФ⁻:

$$f_t = f_{xxx} - 6(f^2 + \lambda)f_x.$$

Определим переменную

$$u = f_x + f^2 + \lambda$$

и проверим, что она удовлетворяет КдФ

$$u_t = u_{xxx} - 6u u_x.$$

Для этого временно поменяем ролями u и f (так как правила дифференцирования у нас заданы для u):

```
In[22]:= ut = u[3] - 6 (u[0]^2 + λ) u[1];
f = u[1] + u[0]^2 + λ;
-star[f, ut] + dx[f, 3] - 6 f dx[f]
Expand[%]
```

```
Out[24]= 12 u[0] u[1]^2 + 6 (λ + u[0]^2) u[2] + 6 u[1] u[2] -
6 (λ + u[0]^2 + u[1]) (2 u[0] u[1] + u[2]) + 2 u[0] u[3] - 2 u[0] (-6 (λ + u[0]^2) u[1] + u[3])
```

```
Out[25]= 0
```

Обращение преобразования Миуры

Мы можем выразить f в виде ряда по обратным степеням параметра $(-\lambda)^{1/2}$, причём коэффициенты ряда будут многочленами от u_n . Сходимость ряда нас не интересует – это просто формальный ряд, производящая функция. Положим $4\lambda = -z^2$ и перепишем преобразование Миуры в виде

$$f^2 + f_x = z^2 / 2 + u.$$

Подставим сюда разложение

$$f = -z/2 + F_0 + F_1/z + F_2/z^2 + \dots$$

(индекс у F_k это просто номер, не производная), это даст

$$z^2/4 - zF_0 + (-F_1 + F_0^2) + (-F_2 + 2F_0F_1)/z + (-F_3 + 2F_0F_2 + F_1^2)/z^2 + \dots + F_{0,x} + F_{1,x}/z + F_{2,x}/z^2 + \dots = z^2/4 + u.$$

Отсюда следует, что $F_0 = 0$, $F_1 = -u$, а на остальные коэффициенты получаем рекуррентные соотношения

$$F_{n+1} = F_1 F_{n-1} + \dots + F_{n-1} F_1 + F_{n,x} = 0, \quad n = 1, 2, 3, \dots$$

Вычислим несколько первых коэффициентов (скажем, 13, хотя машина может гораздо больше).

```
In[26]:= M = 13;
F[1] = -u[0];
Do[F[n + 1] = Factor[(dx[F[n]] + Sum[F[s] F[n - s], {s, 1, n - 1}])], {n, 1, M - 1}];
```

Чтобы было удобнее смотреть, заменим $u[n]$ на u_n :

```
In[29]:= Do[Print[Fn, " = ", F[n] /. u[k_] -> uk], {n, 1, M}]
```

$$F_1 = -u_0$$

$$F_2 = -u_1$$

$$F_3 = u_0^2 - u_2$$

$$F_4 = 4 u_0 u_1 - u_3$$

$$F_5 = -2 u_0^3 + 5 u_1^2 + 6 u_0 u_2 - u_4$$

$$F_6 = -16 u_0^2 u_1 + 18 u_1 u_2 + 8 u_0 u_3 - u_5$$

$$F_7 = 5 u_0^4 - 50 u_0 u_1^2 - 30 u_0^2 u_2 + 19 u_2^2 + 28 u_1 u_3 + 10 u_0 u_4 - u_6$$

$$F_8 = 64 u_0^3 u_1 - 60 u_1^3 - 216 u_0 u_1 u_2 - 48 u_0^2 u_3 + 68 u_2 u_3 + 40 u_1 u_4 + 12 u_0 u_5 - u_7$$

$$F_9 = -14 u_0^5 + 350 u_0^2 u_1^2 + 140 u_0^3 u_2 - 442 u_1^2 u_2 - 266 u_0 u_2^2 - 392 u_0 u_1 u_3 + 69 u_3^2 - 70 u_0^2 u_4 + 110 u_2 u_4 + 54 u_1 u_5 + 14 u_0 u_6 - u_8$$

$$F_{10} = -256 u_0^4 u_1 + 960 u_0 u_1^3 + 1728 u_0^2 u_1 u_2 - 1224 u_1 u_2^2 + 256 u_0^3 u_3 - 900 u_1^2 u_3 - 1088 u_0 u_2 u_3 - 640 u_0 u_1 u_4 + 250 u_3 u_4 - 96 u_0^2 u_5 + 166 u_2 u_5 + 70 u_1 u_6 + 16 u_0 u_7 - u_9$$

$$F_{11} = 42 u_0^6 - 2100 u_0^3 u_1^2 + 1105 u_1^4 - 630 u_0^4 u_2 + 7956 u_0 u_1^2 u_2 + 2394 u_0^2 u_2^2 - 1262 u_2^3 + 3528 u_0^2 u_1 u_3 - 5564 u_1 u_2 u_3 - 1242 u_0 u_3^2 + 420 u_0^3 u_4 - 1630 u_1^2 u_4 - 1980 u_0 u_2 u_4 + 251 u_4^2 - 972 u_0 u_1 u_5 + 418 u_3 u_5 - 126 u_0^2 u_6 + 238 u_2 u_6 + 88 u_1 u_7 + 18 u_0 u_8 - u_{10}$$

$$F_{12} = 1024 u_0^5 u_1 - 9600 u_0^2 u_1^3 - 11520 u_0^3 u_1 u_2 + 13560 u_1^3 u_2 + 24480 u_0 u_1 u_2^2 - 1280 u_0^4 u_3 + 18000 u_0 u_1^2 u_3 + 10880 u_0^2 u_2 u_3 - 9524 u_2^2 u_3 - 7000 u_1 u_3^2 + 6400 u_0^2 u_1 u_4 - 11140 u_1 u_2 u_4 - 5000 u_0 u_3 u_4 + 640 u_0^3 u_5 - 2720 u_1^2 u_5 - 3320 u_0 u_2 u_5 + 922 u_4 u_5 - 1400 u_0 u_1 u_6 + 658 u_3 u_6 - 160 u_0^2 u_7 + 328 u_2 u_7 + 108 u_1 u_8 + 20 u_0 u_9 - u_{11}$$

$$F_{13} = -132 u_0^7 + 11550 u_0^4 u_1^2 - 24310 u_0 u_1^4 + 2772 u_0^5 u_2 - 87516 u_0^2 u_1^2 u_2 - 17556 u_0^3 u_2^2 + 69006 u_1^2 u_2^2 + 27764 u_0 u_2^3 - 25872 u_0^3 u_1 u_3 + 33760 u_1^3 u_3 + 122408 u_0 u_1 u_2 u_3 + 13662 u_0^2 u_3^2 - 26322 u_2 u_3^2 - 2310 u_0^4 u_4 + 35860 u_0 u_1^2 u_4 + 21780 u_0^2 u_2 u_4 - 20922 u_2^2 u_4 - 30776 u_1 u_3 u_4 - 5522 u_0 u_4^2 + 10692 u_0^2 u_1 u_5 - 20376 u_1 u_2 u_5 - 9196 u_0 u_3 u_5 + 923 u_5^2 + 924 u_0^3 u_6 - 4270 u_1^2 u_6 - 5236 u_0 u_2 u_6 + 1582 u_4 u_6 - 1936 u_0 u_1 u_7 + 988 u_3 u_7 - 198 u_0^2 u_8 + 438 u_2 u_8 + 130 u_1 u_9 + 22 u_0 u_{10} - u_{12}$$

Можно проверить, что построенный отрезок ряда действительно удовлетворяет уравнению (5), с точностью до членов z^n с $n \geq M$:

```
In[30]:= f = -z/2 + Sum[F[n]/z^n, {n, 1, M}];
Series[Expand[f^2 + dx[f]], {z, Infinity, M-1}]
```

$$\text{Out[31]} = \frac{z^2}{4} + u[0] + O\left[\frac{1}{z}\right]^{13}$$

Прекрасно, и зачем это было нужно? А вот зачем: оказывается, что многочлены F_n являются *плотностями* законов сохранения для КдФ, то есть, существуют некоторые функции s_n от u_k такие, что

$$D_t(F_n) = D_x(s_n).$$

Эти функции мы пока не знаем, но можем убедиться, что они есть, применив к этому равенству вариационную производную, тогда правая часть уйдёт в 0. Проверяем:

```
In[32]:= Table[
  varD[star[F[n], u[3] - 6 u[0] u[1]],
    {n, 1, M}]
```

$$\text{Out[32]} = \{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\}$$

Если посмотреть внимательно на выписанную выше таблицу, можно заметить, что все F_n с четными номерами сами являются полными производными по x , ещё до дифференцирования по t :

```
In[33]:= Table[vars[F[n]], {n, 2, M, 2}]
```

```
Out[33]= {0, 0, 0, 0, 0, 0}
```

Такие плотности не засчитываются, они отвечают *тривиальным* законам сохранения $D_t(D_x(a)) = D_x(D_t(a))$, которые выполняются для любой функции a и любого эволюционного дифференцирования D_t . А с нечетными номерами все в порядке, это нетривиальные плотности, вариационная производная от них не есть тождественный 0:

```
In[34]:= Table[vars[F[n]], {n, 1, M, 2}]
```

```
Out[34]= {-1, 2 u[0], -6 u[0]^2 + 2 u[2], 20 u[0]^3 - 10 u[1]^2 - 20 u[0] u[2] + 2 u[4],
-70 u[0]^4 + 140 u[0] u[1]^2 + 140 u[0]^2 u[2] - 42 u[2]^2 - 56 u[1] u[3] - 28 u[0] u[4] + 2 u[6],
252 u[0]^5 - 1260 u[0]^2 u[1]^2 - 840 u[0]^3 u[2] + 924 u[1]^2 u[2] + 756 u[0] u[2]^2 + 1008 u[0] u[1] u[3] -
138 u[3]^2 + 252 u[0]^2 u[4] - 228 u[2] u[4] - 108 u[1] u[5] - 36 u[0] u[6] + 2 u[8],
-924 u[0]^6 + 9240 u[0]^3 u[1]^2 - 2310 u[1]^4 + 4620 u[0]^4 u[2] - 20328 u[0] u[1]^2 u[2] -
8316 u[0]^2 u[2]^2 + 2684 u[2]^3 - 11088 u[0]^2 u[1] u[3] + 11352 u[1] u[2] u[3] + 3036 u[0] u[3]^2 -
1848 u[0]^3 u[4] + 3300 u[1]^2 u[4] + 5016 u[0] u[2] u[4] - 506 u[4]^2 + 2376 u[0] u[1] u[5] -
836 u[3] u[5] + 396 u[0]^2 u[6] - 484 u[2] u[6] - 176 u[1] u[7] - 44 u[0] u[8] + 2 u[10]}
```

Чтобы найти соответствующие токи, можно проинтегрировать по частям:

In[35]:=

```
Expand[Table[star[F[n], u[3] - 6 u[0] u[1]], {n, 1, M, 2}]];
int /@ %
```

Out[36]=

```
{ {3 u[0]^2 - u[2], 0}, {-4 u[0]^3 + 5 u[1]^2 + 8 u[0] u[2] - u[4], 0},
{9 u[0]^4 - 60 u[0] u[1]^2 - 42 u[0]^2 u[2] + 19 u[2]^2 + 28 u[1] u[3] + 12 u[0] u[4] - u[6], 0},
{-24 u[0]^5 + 450 u[0]^2 u[1]^2 + 200 u[0]^3 u[2] - 442 u[1]^2 u[2] - 304 u[0] u[2]^2 - 448 u[0] u[1] u[3] +
69 u[3]^2 - 90 u[0]^2 u[4] + 110 u[2] u[4] + 54 u[1] u[5] + 16 u[0] u[6] - u[8], 0},
{70 u[0]^6 - 2800 u[0]^3 u[1]^2 + 1105 u[1]^4 - 910 u[0]^4 u[2] + 8840 u[0] u[1]^2 u[2] +
2926 u[0]^2 u[2]^2 - 1262 u[2]^3 + 4312 u[0]^2 u[1] u[3] - 5564 u[1] u[2] u[3] - 1380 u[0] u[3]^2 +
560 u[0]^3 u[4] - 1630 u[1]^2 u[4] - 2200 u[0] u[2] u[4] + 251 u[4]^2 - 1080 u[0] u[1] u[5] +
418 u[3] u[5] - 154 u[0]^2 u[6] + 238 u[2] u[6] + 88 u[1] u[7] + 20 u[0] u[8] - u[10], 0},
{-216 u[0]^7 + 15750 u[0]^4 u[1]^2 - 26520 u[0] u[1]^4 + 4032 u[0]^5 u[2] - 103428 u[0]^2 u[1]^2 u[2] -
22344 u[0]^3 u[2]^2 + 69006 u[1]^2 u[2]^2 + 30288 u[0] u[2]^3 - 32928 u[0]^3 u[1] u[3] +
33760 u[1]^3 u[3] + 133536 u[0] u[1] u[2] u[3] + 16146 u[0]^2 u[3]^2 - 26322 u[2] u[3]^2 -
3150 u[0]^4 u[4] + 39120 u[0] u[1]^2 u[4] + 25740 u[0]^2 u[2] u[4] - 20922 u[2]^2 u[4] -
30776 u[1] u[3] u[4] - 6024 u[0] u[4]^2 + 12636 u[0]^2 u[1] u[5] - 20376 u[1] u[2] u[5] -
10032 u[0] u[3] u[5] + 923 u[5]^2 + 1176 u[0]^3 u[6] - 4270 u[1]^2 u[6] -
5712 u[0] u[2] u[6] + 1582 u[4] u[6] - 2112 u[0] u[1] u[7] + 988 u[3] u[7] -
234 u[0]^2 u[8] + 438 u[2] u[8] + 130 u[1] u[9] + 24 u[0] u[10] - u[12], 0},
{693 u[0]^8 - 83160 u[0]^5 u[1]^2 + 364650 u[0]^2 u[1]^4 - 17556 u[0]^6 u[2] + 933504 u[0]^3 u[1]^2 u[2] -
496950 u[1]^4 u[2] + 149226 u[0]^4 u[2]^2 - 1932168 u[0] u[1]^2 u[2]^2 - 416460 u[0]^2 u[2]^3 +
174317 u[2]^4 + 219912 u[0]^4 u[1] u[3] - 945280 u[0] u[1]^3 u[3] - 1836120 u[0]^2 u[1] u[2] u[3] +
1533408 u[1] u[2]^2 u[3] - 145728 u[0]^3 u[3]^2 + 562474 u[1]^2 u[3]^2 + 737016 u[0] u[2] u[3]^2 +
16632 u[0]^5 u[4] - 537900 u[0]^2 u[1]^2 u[4] - 232320 u[0]^3 u[2] u[4] + 893724 u[1]^2 u[2] u[4] +
585816 u[0] u[2]^2 u[4] + 861728 u[0] u[1] u[3] u[4] - 206138 u[3]^2 u[4] + 82830 u[0]^2 u[4]^2 -
163722 u[2] u[4]^2 - 114048 u[0]^3 u[1] u[5] + 144780 u[1]^3 u[5] + 570528 u[0] u[1] u[2] u[5] +
137940 u[0]^2 u[3] u[5] - 272108 u[2] u[3] u[5] - 159268 u[1] u[4] u[5] - 25844 u[0] u[5]^2 -
7854 u[0]^4 u[6] + 119560 u[0] u[1]^2 u[6] + 78540 u[0]^2 u[2] u[6] - 76986 u[2]^2 u[6] -
113456 u[1] u[3] u[6] - 44296 u[0] u[4] u[6] + 3431 u[6]^2 + 29040 u[0]^2 u[1] u[7] -
56328 u[1] u[2] u[7] - 27664 u[0] u[3] u[7] + 6004 u[5] u[7] + 2112 u[0]^3 u[8] -
9210 u[1]^2 u[8] - 12264 u[0] u[2] u[8] + 4002 u[4] u[8] - 3640 u[0] u[1] u[9] +
2000 u[3] u[9] - 330 u[0]^2 u[10] + 726 u[2] u[10] + 180 u[1] u[11] + 28 u[0] u[12] - u[14], 0}}
```

Вторые компоненты в парах равны 0, значит первые компоненты дают токи.

Итак, можно сформулировать такие утверждения (доказательство в лекции).

Теорема. 1) Все коэффициенты в разложении (6) являются плотностями законов сохранения для КдФ;
2) плотности с чётными номерами тривиальны; 3) плотности с нечётными номерами нетривиальны.